The Stochastic Power Law Logistic Model: Derivations of Asymptotic Approximations of the first 3 Cumulants of the QSD, with s=6.

Ingemar Nåsell

The procedure **cumf** is used to determine the cumulant of order kk as a function of the (raw) moments of orders up to kk, while the procedure **muf** determines the (raw) moment of order kk as a function of the cumulants of orders up to kk. The procedure **ODEcum3** derives ODEs for the first 3 cumulants.
More details are given in the Maple work-sheet DeriveODECum3.

```
> restart;
```

```
> cumf:=proc(kk)
   local M,K,Ks,cum;
   description "Determines the cumulant of order kk as function of
  the raw moments of orders up to kk";
   M:=1+add(cat(mu,k)*theta^k/k!,k=1..kk);
   K:=log(M);
   Ks:=convert(series(K,theta,kk+1),polynom);
   cum:=sort(simplify(coeff(Ks,theta,kk)*kk!),[seq(cat(mu,j),j=1..
  kk)],plex);
  end proc:
```

```
> muf:=proc(kk)
   local K,M,Ms,mu;
   description "Determines the (raw) moment of order kk as function
  of the cumulants of orders up to kk";
   K:=add(cat(kappa,k)*theta^k/k!,k=1..kk);
   M:=exp(K);
   Ms:=convert(series(M,theta,kk+1),polynom);
   mu:=sort(coeff(Ms,theta,kk)*kk!,[seq(cat(kappa,j),j=1..kk)],
  plex);
  end proc:
```

```
> ODEcum3:=proc(s)
    local Dmu1,Dmu2,Dmu3,sub1,sub2,Dcum1,Dcum2,Dcum3,a,c,d,b;
    description "Derive ODEs for the first 3 cumulants";
    Dmu1:=a*mu1-b*cat(mu,s+1);
    Dmu2:=2*a*mu2+c*mu1-d*cat(mu,s+1)-2*b*cat(mu,s+2);
    Dmu3:=a*(mu1+3*mu3)+3*c*mu2-3*d*cat(mu,s+2)-b*(cat(mu,s+1)+3*
  cat(mu,s+3));
    sub1:=D(mu1)=Dmu1,D(mu2)=Dmu2,D(mu3)=Dmu3;
    sub2:=seq(cat(mu,i)=muf(i),i=1..3+s);
    Dcum1:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
  (1)))),[a,c,d,b])));
```

```
  Dcum2:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
(2)))),[a,c,d,b])));
  Dcum3:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
(3)))),[a,c,d,b])));
  a:=mu*(R0-1);
  c:=mu*(R0+1);
  d:=mu*(R0-alpha)/N^s;
  b:=mu*(R0+alpha)/N^s;
  [eval(Dcum1),eval(Dcum2),eval(Dcum3)];
end proc:
```

Put s=6 and denote the derivatives of the first 3 cumulants by A, B, C.

```
> s:=6;
```

$$s := 6 \tag{1}$$

```
> A:=ODEcum3(s)[1];
  B:=ODEcum3(s)[2];
  C:=ODEcum3(s)[3];
```

$$A := \mu\,(R0-1)\,\kappa 1 - \frac{1}{N^6}\Big(\big(\kappa 1^7 + 21\,\kappa 1^5\,\kappa 2 + 35\,\kappa 1^4\,\kappa 3 + \big(105\,\kappa 2^2 + 35\,\kappa 4\big)\,\kappa 1^3 + 105\,\kappa 2^2\,\kappa 3 + (210\,\kappa 2\,\kappa 3 + 21\,\kappa 5)\,\kappa 1^2 + 21\,\kappa 2\,\kappa 5 + 35\,\kappa 3\,\kappa 4 + \big(105\,\kappa 2^3 + 105\,\kappa 2\,\kappa 4 + 70\,\kappa 3^2 + 7\,\kappa 6\big)\,\kappa 1 + \kappa 7\big)\,\mu\,(R0+\alpha)\Big)$$

$$B := 2\,\mu\,(R0-1)\,\kappa 2 + \mu\,(R0+1)\,\kappa 1 - \frac{1}{N^6}\Big(\big(\kappa 1^7 + 21\,\kappa 1^5\,\kappa 2 + 35\,\kappa 1^4\,\kappa 3 + \big(105\,\kappa 2^2 + 35\,\kappa 4\big)\,\kappa 1^3 + 105\,\kappa 2^2\,\kappa 3 + (210\,\kappa 2\,\kappa 3 + 21\,\kappa 5)\,\kappa 1^2 + 21\,\kappa 2\,\kappa 5 + 35\,\kappa 3\,\kappa 4 + \big(105\,\kappa 2^3 + 105\,\kappa 2\,\kappa 4 + 70\,\kappa 3^2 + 7\,\kappa 6\big)\,\kappa 1 + \kappa 7\big)\,\mu\,(R0-\alpha)\Big) - \frac{1}{N^6}\left(14\left(\kappa 1^6\,\kappa 2 + 3\,\kappa 1^5\,\kappa 3 + \big(15\,\kappa 2^2 + 5\,\kappa 4\big)\,\kappa 1^4 + 15\,\kappa 2^4 + (50\,\kappa 2\,\kappa 3 + 5\,\kappa 5)\,\kappa 1^3 + 30\,\kappa 2^2\,\kappa 4 + \big(45\,\kappa 2^3 + 45\,\kappa 2\,\kappa 4 + 30\,\kappa 3^2 + 3\,\kappa 6\big)\,\kappa 1^2 + 8\,\kappa 3\,\kappa 5 + 5\,\kappa 4^2 + \big(105\,\kappa 2^2\,\kappa 3 + 21\,\kappa 2\,\kappa 5 + 35\,\kappa 3\,\kappa 4 + \kappa 7\big)\,\kappa 1 + \big(40\,\kappa 3^2 + 4\,\kappa 6\big)\,\kappa 2 + \frac{\kappa 8}{7}\right)\mu\,(R0+\alpha)\right)$$

$$C := (\kappa 1 + 3\,\kappa 3)\,\mu\,(R0-1) + 3\,\mu\,(R0+1)\,\kappa 2 - \frac{1}{N^6}\left(21\left(\kappa 1^6\,\kappa 2 + 3\,\kappa 1^5\,\kappa 3 + \big(15\,\kappa 2^2 + 5\,\kappa 4\big)\,\kappa 1^4 + 15\,\kappa 2^4 + (50\,\kappa 2\,\kappa 3 + 5\,\kappa 5)\,\kappa 1^3 + 30\,\kappa 2^2\,\kappa 4 + \big(45\,\kappa 2^3 + 45\,\kappa 2\,\kappa 4 + 30\,\kappa 3^2 + 3\,\kappa 6\big)\,\kappa 1^2 + 8\,\kappa 3\,\kappa 5 + 5\,\kappa 4^2 + \big(105\,\kappa 2^2\,\kappa 3 + 21\,\kappa 2\,\kappa 5 + 35\,\kappa 3\,\kappa 4 + \kappa 7\big)\,\kappa 1 + \big(40\,\kappa 3^2 + 4\,\kappa 6\big)\,\kappa 2 + \frac{\kappa 8}{7}\right)\mu\,(R0-\alpha)\right) \tag{2}$$

$$-\frac{1}{N^6}\left(3\left(\frac{\kappa1^7}{3}+7\,\kappa1^6\,\kappa3+\left(42\,\kappa2^2+7\,\kappa2+21\,\kappa4\right)\kappa1^5+\left(315\,\kappa2\,\kappa3\right.\right.\right.$$
$$+\frac{35}{3}\,\kappa3+35\,\kappa5\Big)\,\kappa1^4+1155\,\kappa2^3\,\kappa3+\Big(420\,\kappa2^3+35\,\kappa2^2+490\,\kappa2\,\kappa4$$
$$+350\,\kappa3^2+\frac{35}{3}\,\kappa4+35\,\kappa6\Big)\,\kappa1^3+280\,\kappa3^3+\big(1995\,\kappa2^2\,\kappa3+735\,\kappa3\,\kappa4$$
$$+(70\,\kappa3+420\,\kappa5)\,\kappa2+7\,\kappa5+21\,\kappa7\big)\,\kappa1^2+(35\,\kappa3+357\,\kappa5)\,\kappa2^2$$
$$+126\,\kappa4\,\kappa5+\Big(630\,\kappa2^4+35\,\kappa2^3+1365\,\kappa2^2\,\kappa4+\frac{70\,\kappa3^2}{3}+392\,\kappa3\,\kappa5$$
$$+245\,\kappa4^2+\big(1890\,\kappa3^2+35\,\kappa4+189\,\kappa6\big)\,\kappa2+\frac{7\,\kappa6}{3}+7\,\kappa8\Big)\,\kappa1$$
$$+(1225\,\kappa3\,\kappa4+7\,\kappa5+35\,\kappa7)\,\kappa2+\left(\frac{35\,\kappa4}{3}+84\,\kappa6\right)\kappa3+\frac{\kappa7}{3}+\kappa9\Big)\,\mu\,(R0$$
$$\left.+\alpha)\right)$$

Determine asymptotic approximations of the first 3 cumulants of the following forms. Assume also that kappa4 - kappa9 are O(N).

```
> kappa1:=x1*N + x2 + x3/N:
  kappa2:=y1*N + y2:
  kappa3:=z1*N:
  kappa4:=u1*N:
  kappa5:=u2*N:
  kappa6:=u3*N:
  kappa7:=u4*N:
  kappa8:=u5*N:
  kappa9:=u6*N:
```

Asymptotic expressions for A, B, C are written A=A1*N + A2 + A3/N, B=B1*N + B2, C =C1*N, where

```
> A1:=coeff(A,N,1); A2:=coeff(A,N,0); A3:=coeff(A,N,-1);
  B1:=coeff(B,N,1); B2:=coeff(B,N,0);
  C1:=coeff(C,N,1);
```

$$A1 := \mu\,(R0-1)\,x1 - x1^7\,\mu\,(R0+\alpha)$$

$$A2 := \mu\,(R0-1)\,x2 - \left(7\,x2\,x1^6+21\,x1^5\,y1\right)\mu\,(R0+\alpha)$$

$$A3 := \mu\,(R0-1)\,x3 - \Big(x3\,x1^6+6\,x2^2\,x1^5+x1\,\big(2\,(x3\,x1^2+2\,x2^2\,x1+x1\,(2\,x3\,x1$$
$$+x2^2))\big)\,x1^3+9\,x2^2\,x1^4\big)+21\,x1^5\,y2+105\,x2\,x1^4\,y1+35\,x1^4\,z1$$
$$+105\,y1^2\,x1^3\Big)\,\mu\,(R0+\alpha)$$

$$B1 := 2\,\mu\,(R0-1)\,y1+\mu\,(R0+1)\,x1-x1^7\,\mu\,(R0-\alpha)-14\,x1^6\,y1\,\mu\,(R0+\alpha)$$

$$B2 := 2\,\mu\,(R0-1)\,y2+\mu\,(R0+1)\,x2-\left(7\,x2\,x1^6+21\,x1^5\,y1\right)\mu\,(R0-\alpha)$$

$$-14\,(x1^6\,y2+6\,x2\,x1^5\,y1+3\,x1^5\,z1+15\,y1^2\,x1^4)\,\mu\,(R0+\alpha)$$

$$C1 := (x1+3\,z1)\,\mu\,(R0-1)+3\,\mu\,(R0+1)\,y1-21\,x1^6\,y1\,\mu\,(R0-\alpha)-3\left(\frac{1}{3}\,x1^7+7\,x1^6\,z1+42\,y1^2\,x1^5\right)\mu\,(R0+\alpha) \quad (3)$$

6 equations can now be formed by setting each of A1, A2, A3, B1, B2, C1 equal to zero. These equations can be solved for the 6 unknowns x1, x2, x3, y1, y2, z1. The solutions are found sequentially, as follows:
First we solve A1=0 for x1.This equation has s+1=7 solutions. Among them, we exclude s=6 spurious solution.
After this, B1=0 is solved for for y1, A2=0 is solved for x2, C1=0 is solved for z1, B2=0 is solved for y2, and A3=0 is solved for x3.

```
> x1a:=solve(A1,x1);
  x1:=((R0-1)/(R0+alpha))^(1/6);
```

$$x1a := 0,\ \frac{((R0-1)\,(R0+\alpha)^5)^{1/6}}{R0+\alpha},\ \frac{\left(\frac{1}{2}+\frac{I\sqrt{3}}{2}\right)((R0-1)\,(R0+\alpha)^5)^{1/6}}{R0+\alpha},\ \frac{\left(-\frac{1}{2}+\frac{I\sqrt{3}}{2}\right)((R0-1)\,(R0+\alpha)^5)^{1/6}}{R0+\alpha},\ -\frac{((R0-1)\,(R0+\alpha)^5)^{1/6}}{R0+\alpha},\ \frac{\left(-\frac{1}{2}-\frac{I\sqrt{3}}{2}\right)((R0-1)\,(R0+\alpha)^5)^{1/6}}{R0+\alpha},\ \frac{\left(\frac{1}{2}-\frac{I\sqrt{3}}{2}\right)((R0-1)\,(R0+\alpha)^5)^{1/6}}{R0+\alpha}$$

$$x1 := \left(\frac{R0-1}{R0+\alpha}\right)^{1/6} \quad (4)$$

```
> y1:=factor(solve(B1,y1));
```

$$y1 := \frac{\left(\frac{R0-1}{R0+\alpha}\right)^{1/6} R0\,(\alpha+1)}{6\,(R0+\alpha)\,(R0-1)} \quad (5)$$

```
> x2:=solve(A2,x2);
```

$$x2 := -\frac{7\,R0\,(\alpha+1)}{12\,(R0+\alpha)\,(R0-1)} \quad (6)$$

```
> z1:=factor(solve(C1,z1));
```

$$z1 := -\frac{\left(\frac{R0-1}{R0+\alpha}\right)^{1/6} R0\,(\alpha+1)\,(6\,R0^2-R0\,\alpha-R0+6\,\alpha)}{36\,(R0+\alpha)^2\,(R0-1)^2} \quad (7)$$

```
> y2:=solve(B2,y2);
```

$$y2 := \frac{7\,(R0^2+\alpha)\,R0\,(\alpha+1)}{12\,(R0+\alpha)^2\,(R0-1)^2} \tag{8}$$

```
> x3:=factor(solve(A3,x3));
```

$$x3 := -\frac{7\,R0\,(\alpha+1)\,(132\,R0^2+17\,R0\,\alpha+17\,R0+132\,\alpha)}{864\left(\frac{R0-1}{R0+\alpha}\right)^{1/6}(R0-1)^2\,(R0+\alpha)^2} \tag{9}$$

Summarize the results for s=6:

```
> x1, x2, x3, y1, y2, z1;
```

$$\left(\frac{R0-1}{R0+\alpha}\right)^{1/6}, -\frac{7\,R0\,(\alpha+1)}{12\,(R0+\alpha)\,(R0-1)},$$
$$-\frac{7\,R0\,(\alpha+1)\,(132\,R0^2+17\,R0\,\alpha+17\,R0+132\,\alpha)}{864\left(\frac{R0-1}{R0+\alpha}\right)^{1/6}(R0-1)^2\,(R0+\alpha)^2},$$
$$\frac{\left(\frac{R0-1}{R0+\alpha}\right)^{1/6}R0\,(\alpha+1)}{6\,(R0+\alpha)\,(R0-1)}, \frac{7\,(R0^2+\alpha)\,R0\,(\alpha+1)}{12\,(R0+\alpha)^2\,(R0-1)^2},$$
$$-\frac{\left(\frac{R0-1}{R0+\alpha}\right)^{1/6}R0\,(\alpha+1)\,(6\,R0^2-R0\,\alpha-R0+6\,\alpha)}{36\,(R0+\alpha)^2\,(R0-1)^2} \tag{10}$$

```
>
```